

Drug Dosages: Are They Effective? Are They Safe?

Note: You may notice differences between this Maple worksheet and the equivalent Mathematica notebook. These differences were introduced to preserve the content of these modules and were necessary because of major functional differences between Maple and Mathematica.

Introduction

OBJECTIVE: Use *Maple* to solve differential equations related to drug concentrations in the blood when the drug is administered by a single injection, intravenously, or by periodic injections.

You will learn how a simple mathematical model and *Maple* can be used to regulate the concentration of a prescribed drug in the blood. You will use *Maple's* differential equation solver to obtain solutions for the drug concentration in the blood when the drug is administered by a single injection, intravenously, or by a series of periodic injections. You will also develop drug dosage regimens to keep the concentration of the drug within the safe and effective concentration limits.

Technology Guidelines

NOTE: If you have just finished a worksheet, **restart** *Maple* before executing a new worksheet.
TO OPEN SECTIONS,

Click on the **PLUS** sign at the left hand side of the screen *or* select **Expand All Sections** from the **View** drop down menu.

TO STOP AN EXECUTION

Click on **STOP** button from the toolbar.

ORDER OF EXECUTION

Execute commands in the order given. Do not skip any *Maple* Input lines within a given worksheet

Alternatively, you can execute the entire worksheet by selecting the **Execute Worksheet** command from the **Edit** drop down menu.

SAVING WORKSHEETS.

You can save anytime to any directory you choose, and it is wise to save often.

EXPERIENCING MAJOR PROBLEMS

Save if appropriate, and then shut down *Maple* and start it up again.

Part I: A Single Dose

An Initial Value Problem and its Solution

When a certain drug is administered by injection, the body absorbs it quickly so that the concentration of the drug in the blood rises to a peak value shortly after the drug is given. At the same time, the kidneys work to remove the drug from the blood, and the rate of removal is proportional to drug concentration in the blood. This process is modeled by the

initial value problem $\frac{dc}{dt} = -kc$ with $c(0) = c_0$, where $c(t)$ is the concentration of the

drug in the blood at time t , and c_0 is the initial concentration when the injection is given at time $t = 0$. This model assumes that at the time of the injection, the drug is immediately absorbed into the blood.

Even though we know that $c(t) = c_0 e^{(-kt)}$ is the solution of the initial value problem,

we will use *Maple's* **dsolve()** command to obtain this solution.

```
> restart;
  soln := dsolve( {diff(c(t),t) = -k*c(t), c(0) = c0}, c(t) );
```

$$\text{soln} := c(t) = c_0 e^{(-kt)}$$

Now we can form a function, **c(t)**, by using the rule from the solution of the initial value problem to substitute the solution $c_0 e^{(-kt)}$ for the expression

c(t).

```
> c := unapply( rhs( soln ), t );
```

$$c := t \rightarrow c_0 e^{(-kt)}$$

The last command effectively replaces **c(t)** with $c_0 e^{(-kt)}$ and assigns it to a redefined function **c**.

Determining k and Plotting Concentrations

One hour after the drug is injected, a blood test from a particular patient shows that the drug

concentration in the blood drops to 80% of its initial concentration. From this information, we can determine the value of k for this patient. We do this by setting $c(t)$ equal to $.80 c_0$ at $t = 1$ hour, and solving for k .

```
> ksoln := solve( c(1) = 0.8 * c0, {k} );
```

```
ksoln := { k = 0.2231435513 }
```

```
> assign(ksoln);
```


Let's redefine $c(t)$, substituting the value for k .

```
> c := unapply( rhs( soln ), t );
```

```
c := t → c0 e(-0.2231435513 t)
```

Suppose an injection of the drug raises the concentration to 3 milligrams per liter, and then the concentration decays as time progresses. Let's plot the solution, substituting 3 mg/l for c_0 .


```
> plot( eval( c(t), c0=3 ), t=0..24, labels=["t(hr)", "c(mg/L)"], color=blue, labeldirection [HORIZONTAL,VERTICAL]);
```



How Long is the Drug Effective?

This particular drug is effective only when the drug concentration in the blood is above 1 mg/l. Let's plot the drug concentration together with the effective threshold concentration.

```
> plot( [eval( c(t), c0=3 ), 1], t=0..24, labels=["t(hr)", "c(mg/L)"] , color=[blue, green],labeldirections=[HORIZONTAL,VERTICAL]);
```



>From the graph, can you estimate how long after the injection the drug will be effective for this patient?

We can determine exactly how long after an injection the drug will be effective, assuming no other injections are given. We do this by finding the time when c reaches 1 mg/l.

```
> teff := solve( eval( c(t), c0=3 ) = 1, {t} );
```

```
teff := {t = 4.923343212}
```

This solution indicates that the drug will be effective for about 4 hours and 55 minutes after the patient receives an injection that boosts the drug concentration in the blood to 3 mg/l.

You Try It: On Another Patient

Another patient needs the same drug. A blood test on this new patient shows that after 1 hour, the drug concentration in the blood drops to 72% of the concentration right after the injection was given.

1. Determine the function that gives the drug concentration in the blood, t hours after the injection is given. To assist you, we have copied the commands you need in the following cell. Change the

value of **n** in the next cell.

```
> restart;
n:=0.80:
soln := dsolve( {D(c)(t) = -k*c(t), c(0) = c0}, c(t) ):
c := unapply( rhs( soln ), t ):
ksoln := solve( c(1) = n * c0, {k} ):
map( assign, ksoln ):
c := unapply( rhs( soln ), t );
```

$$c := t \rightarrow c0 e^{(-0.2231435513 t)}$$


2. For the new patient, modify the command in the next cell to determine the initial drug concentration, **c0**, so that the drug will remain effective for 5 hours. Assume that the effectiveness threshold is 1 mg/l.

```
> teff := solve( eval( c(t), c0=3 ) = 1, {t} );
```

$$teff := \{t = 4.923343212\}$$

Plot **c(t)** using the value for **c0** that you found in step 2.

```
> plot( [eval( c(t), c0=3 ), 1], t=0..24, labels=["t(hr)", "c(mg/L)"], color=[blue, green],
labeldirections=[HORIZONTAL,VERTICAL]);
```



Part II: Intravenous Dosage

Sometimes it is desirable to maintain a constant concentration of the drug in the blood. This can be accomplished with intravenous (IV) injection. In this case, the drug is continuously fed into the bloodstream at a constant rate. We will modify the initial value problem in Part I to model intravenous administration of the drug.

A Model for Intravenous Dosage

The rate of change of the drug concentration in the blood is the difference between the rate at which the drug concentration increases in the blood due to the IV, and the rate at which it is removed by the kidneys. That is, $\frac{dc}{dt} = (\text{rate of increase to IV} - \text{rate of decrease due to}$

kidneys). We assume that for IV administration of the drug, the rate of increase in the concentration of the drug due to the IV is constant and, as before, that the rate of removal is proportional to the concentration. Therefore, $\frac{dc}{dt} = r - kc$, where r is the constant rate of

increase in the drug concentration due to the IV and kc is the rate at which the kidneys reduce the concentration. For IV administration, we will assume that the initial concentration is 0, that is, $c(0) = 0$. Now we can solve the initial value problem.

```
> restart;
    soln := dsolve( {diff(c(t),t) = r-k*c(t), c(0) = 0}, c(t) );
```

$$\text{soln} := c(t) = \frac{r}{k} - \frac{e^{(-k t)} r}{k}$$

```
> c := unapply( rhs( soln ), t );
```

$$c := t \rightarrow \frac{r}{k} - \frac{e^{(-k t)} r}{k}$$

The Steady-State Concentration, $c(\infty)$

What does $c(t)$ approach as $t \rightarrow \infty$, if k is a positive constant? The limiting value is often called the steady state or equilibrium solution and is denoted by $c(\infty)$. You can also


obtain this value by setting $\frac{dc}{dt} = 0$ in the differential equation and solving for c . What do

you get for the steady-state solution when you find it the second way? Is it the same result

that you obtained by taking the limit of $c(t)$ as $t \rightarrow \infty$?

Assuming that the patient is the same person as in Part I, so that the k value is the same, that is, $k = 0.223144$, and assuming that the rate of increase in the drug concentration due to the IV is $1.0 \frac{\text{mg/l}}{\text{hour}}$, we can plot the solution for $c(t)$, substituting these values for r and k .

> **plot([eval(c(t), [r=1.0, k=0.223144]), 1], t=0..24, labels=["t(hours)", "c(mg/L)"], col [blue,green], labeldirections=[HORIZONTAL,VERTICAL]);**



Using **eval(c(t), [r=1.0, k=0.223144]** to substitute values for the parameters r and k , as we did in the preceding plot command, is useful when we want to preserve the general form of $c(t)$. This way we can substitute other values into the general solution without having to redefine the function $c(t)$. We check to see that the general form of $c(t)$ is preserved.

> **c(t);**

$$\frac{r}{k} - \frac{e^{(-k t)} r}{k}$$

How Long Does it Take to be Effective?

Next, we determine how long it takes the drug concentration in the blood to reach the effective threshold value of 1 mg/l.

```
> solve( eval( c(t), [r=1.0, k=0.223144] ) = 1, {t} );
```

```
{t = 1.131557532}
```

It takes over an hour for the drug to become effective. One way to reduce this time is to administer a loading dose by injection at $t = 0$. This instantly brings the drug concentration in the blood to 1mg/l, and then the IV is started. The differential equation is the same as for IV administration without a loading dose, but the initial condition is different.

```
> unassign('c,k,soln');
```

```
> soln := dsolve( {D(c)(t) = r - k*c(t), c(0) = 1}, c(t) );
```


$$\text{soln} := c(t) = \frac{r}{k} + e^{(-k t)} \left(1 - \frac{r}{k} \right)$$

```
> c := unapply( rhs( soln ), t );
```

$$c := t \rightarrow \frac{r}{k} + e^{(-k t)} \left(1 - \frac{r}{k} \right)$$

Let's plot $c(t)$.

```
> plot( [eval( c(t), [r=1.0, k=0.223144] ), 1], t=0..24, labels=["t(hours)", "c(mg/L)"], view=0..5, color=[blue,green], labeldirections=[HORIZONTAL,VERTICAL]);
```



You Try It: Is the Dosage Safe?

The drug is effective provided its concentration in the blood is above 1 mg/l, however, if the concentration exceeds 3 mg/l, the patient will have serious side effects from the drug. Consequently, the dosage prescriptions in Part II are unsafe for the patient.

1. Modify the drug dosage prescription so that the concentration is always above the effective threshold and the steady-state concentration is halfway between the safe and effective threshold values, and plot $c(t)$ over 24 hours. First use trial and error, and then use the analytic solution for the steady-state concentration to determine the correct value for r . To assist you we have copied the commands that you need in the following input cell. Change **rvalue**.

```
> unassign('c,k,soln');

> rvalue := 1.0;
   kvalue := 0.223144;
   soln := dsolve( {D(c)(t) = r - k*c(t), c(0) = 1}, c(t) );
   c := unapply( rhs(soln), t );
   printf( `The drug concentration is %a`, eval( c(t), [r=rvalue, k=kvalue] ) );
   printf( `The steady-state concentration is %a`, rvalue/kvalue );
   plot([eval( c(t), [r=rvalue, k=kvalue] ), 1, 3], t=0..24, labels=["t(hourse)", "c(mg/L)"], view:
[0..24, 0..5] , color=[blue, green, red], labeldirections=[HORIZONTAL,VERTICAL]);
```

$$rvalue := 1.0$$


$$kvalue := 0.223144$$

$$soln := c(t) = \frac{r}{k} + e^{(-k t)} \left(1 - \frac{r}{k} \right)$$

$$c := t \rightarrow \frac{r}{k} + e^{(-k t)} \left(1 - \frac{r}{k} \right)$$

The drug concentration is $4.481411107 - 3.481411107 \cdot \exp(-.223144 \cdot t)$

The steady-state concentration is 4.481411107



2. Repeat exercise 1 for the other patient from You Try It: Part I. Change **rvalue** and **kvalue**.

Part III: Maintenance Dosages at Discrete Time Intervals

An alternative method for administering the drug is to give periodic injections to maintain a safe and effective drug concentration in the blood. Again, we assume that the drug is completely absorbed as soon as it is administered and that the kidneys reduce the drug concentration at a rate proportional to the concentration c . The differential equation model is $\frac{dc}{dt} = -kc + d(t)$, where d

(t) is the rate of increase in the drug concentration due to administration of the drug. The initial condition is the concentration of the drug in the blood at $t = 0$. If there is no loading dosage, then $c(0) = 0$, but if a loading dose of c_0 is given, then $c(0) = c_0$.

First, we specify a maintenance dosage that we will administer periodically.

```
> restart;
    maintenancedose := 2.0;
```

```
maintenancedose := 2.0
```

Next, we use a special mathematical function called the Dirac delta to model the periodic injections. The Dirac delta allows us to increase the drug concentration in the blood by a finite amount at specified times. In the next cell, we show how to use the Dirac delta to model giving a maintenance dosage at $t = 4$ hours and then again at $t = 8$ hours.

```
> dosageschedule := maintenancedose*Dirac(t - 4) + maintenancedose*Dirac(t - 8);
```

$$\text{dosageschedule} := 2.0 \operatorname{Dirac}(t - 4) + 2.0 \operatorname{Dirac}(t - 8)$$

Now we solve the initial value problem to determine the drug concentration in the blood as a function of time. We will take the loading dosage to be 2.0, the same as the maintenance dosage, and replace $d(t)$ in the differential equation with **dosageschedule**.

```
> kvalue := 0.223144:
soln := dsolve( {D(c)(t) = -k*c(t)+dosageschedule, c(0) = maintenancedose}, c(t) );
```

$$\text{soln} := c(t) = (2 e^{(4 k)} \operatorname{Heaviside}(t - 4) + 2 e^{(8 k)} \operatorname{Heaviside}(t - 8) + 2) e^{(-k t)}$$


```
> c := unapply( rhs(soln), t );
```

$$c := t \rightarrow (2 e^{(4 k)} \operatorname{Heaviside}(t - 4) + 2 e^{(8 k)} \operatorname{Heaviside}(t - 8) + 2) e^{(-k t)}$$

The Heaviside function, also known as the unit step function, that appears in the solution is another special mathematical function. It acts like an "on-off" switch. The function $\operatorname{Heaviside}(t, a)$ has a value of 0 when $t < a$, and a value of 1 when $a \leq t$.

Next, we plot the solution, substituting the appropriate value for **k**.


```
> plot([eval( c(t), [k=kvalue] ), 1, 3], t=0..24, labels=["t(hours)", "c(mg/L)"], color=[blue,green,red],
labeldirections=[HORIZONTAL,VERTICAL]);
```



Now we use the Dirac delta and the **sum()** command to build a dosage maintenance schedule to

use as an input function for the differential equation. It is apparent that the previous schedule fell outside the safe and effective range, so we will try something different. We will try 2.5 mg/l every 6 hours.

```
> restart;
maintenancedose := 2.5;
deltat := 6;
kvalue := 0.223144;
dosageschedule := sum( maintenancedose*Dirac(t - n*deltat), n=1..24/deltat );
soln := dsolve( {D(c)(t) = -k*c(t)+dosageschedule, c(0) = maintenancedose}, c(t) );
c := unapply( rhs(soln), t );
plot([eval( c(t), [k=kvalue] ), 1, 3], t=0..24, labels=["t(hours)", "c(mg/L)"], color=[blue,green],labeldirections=[HORIZONTAL,VERTICAL]);
```



Well, that didn't work either.

You Try It: Safe and Effective Drug Dosage


1. Try to build a drug dosage schedule that keeps the drug concentration in the safe and effective range, that is, between 1.0 and 3.0, and that maximizes patient comfort at the same time. We have copied the command that you will need in the next cell. Change the items that are **maintenancedose** and **deltat**.

```
> restart;
maintenancedose := 2.5;
deltat := 6;
kvalue := 0.223144;
dosageschedule := sum( maintenancedose*Dirac(t - n*deltat), n=1..24/deltat );
soln := dsolve( {D(c)(t) = -k*c(t)+dosageschedule, c(0) = maintenancedose}, c(t) );
```

```
c := unapply( rhs(soln), t );  
plot( [eval( c(t), [k=kvalue] ), 1, 3], t=0..24, labels=["t(hours)", "c(mg/L)"], color=[blue,gr  
red], labeldirections=[HORIZONTAL,VERTICAL]);
```

```
dosageschedule := 2.500000000 Dirac(t - 6.) + 2.500000000 Dirac(t - 12.) + 2.500000000 Dirac(t - 18.)  
+ 2.500000000 Dirac(t - 24.)
```

$$c := t \rightarrow \left(\frac{5}{2} e^{(24k)} \text{Heaviside}(t - 24) + \frac{5}{2} e^{(6k)} \text{Heaviside}(t - 6) + \frac{5}{2} e^{(12k)} \text{Heaviside}(t - 12) \right. \\ \left. + \frac{5}{2} e^{(18k)} \text{Heaviside}(t - 18) + \frac{5}{2} \right) e^{(-kt)}$$



>